Population Pharmacokinetic Modeling of Oxcarbazepine and its Active Metabolite 10-Monohydroxy Derivative to Inform Dosing in Children with Obesity

Electronic Supplementary Material 2

Clinical Pharmacokinetics

Jaydeep Sinha, B.Pharm, M.Pharm, PhD[1], Kanecia Zimmerman, MD, PhD, MPH[2,3], Stephen J. Balevic, MD, PhD, MPH[2,3,4], Chi Hornik, PharmD[2,3], William J. Muller, MD, PhD[5], Mobeen Rathore, MD[6], Marisa Meyer, DO[7,8]; Yaron Finkelstein, MD[9], Amira Al-Uzri, MD[10], Arpita Lakhotia, MBBS[11]; Stuart Goldstein, MD[12]; Jia-Yuh Chen, PhD[13], Ravinder Anand, MS, PhD[13], and Daniel Gonzalez, PharmD, PhD[3,4] on behalf of the Best Pharmaceuticals for Children Act – Pediatric Trials Network Steering Committee

[1]Department of Pediatrics, UNC School of Medicine, The University of North Carolina at Chapel Hill, Chapel Hill, NC, USA (at the time of performing this work); [2]Department of Pediatrics, Duke University School of Medicine, Durham, NC, USA; [3]Duke Clinical Research Institute, Durham, NC, USA; [4]Department of Medicine, Duke University School of Medicine, Durham, NC, USA; [5]Ann & Robert H. Lurie Children's Hospital of Chicago, Chicago, IL, USA; [6]University of Florida Center for HIV/AIDS Research, Education and Service (UF CARES), Jacksonville Shands Medical Center, Jacksonville, FL, US; [7]Division of Critical Care, Department of Pediatrics, Nemours Children's Hospital, Wilmington, DE, USA; [8]Department of Pediatrics, Sidney Kimmel Medical College at Thomas Jefferson University, Philadelphia, PA, USA; [9]The Hospital for Sick Children, Toronto, Ontario, Canada; [10]Oregon Health and Science

University, Portland, OR, USA; [11]University of Louisville Norton Children's Medical Group, Louisville, KY, USA; [12]Cincinnati Children's Hospital Medical Center, Cincinnati, OH, USA; [13]The Emmes Company, LLC, Rockville, MD, USA.

Address correspondence to: Daniel Gonzalez, Duke Clinical Research Institute, PO Box 17969, Durham, NC, 27715; daniel.gonzalez@duke.edu.

SUPPLEMENTARY MATERIALS 2

CONTENTS

NONMEM control stream for the final model (M6)

```
;Description: Joint One Comp Model With Back Transformation
;Author: Jaydeep Sinha

$PROBLEM    Parent-metabolite model of oxcarbazepine (OXZ) and 10-monohydroxy
derivative (MHD) using AED01 and POP01 merged dataset

;Units: AMT=mg, TAFD=h, TALD=h, Sex = 1(male) or 2(female), WTKG=kg, PKWT= kg,
;HTCM=cm, PNA=year, PMADY=day, PMAWK=week, BMI=kg/m2, SCR=mg/dL, BUN=mg/dL,
;AST=U/L, ALT=U/L, ALB=g/dL, PROT=g/dL, CLCR= mL/min/1.73 m2

$INPUT     STUDY SITENUM ID DATE=DROP TIME EVID DOSN AMT CMT CMT2=DROP
DRTE DFRM TYPE DV MDV TAFD TALD SEX ETHN RACE RACEWB WTKG PKWT
WTDT=DROP HTCM AGRP PNA PMADY PMAWK OBESE BMI PBMI SCR BUN AST ALT
ALB PROT KETO CARB VERA PHENOB VALP TOPR LEVE PHENYT FOSPHENYT CLCR
L2=DROP

$DATA      oxz_mhd_v9.csv IGNORE=#

$SUBROUTINE ADVAN6 TOL=4
$MODEL NCOMP=3

COMP = (DEPOT1)                  ; (1) OXZ dose
COMP = (CENTRAL1)                ; (2) OXZ central (observation-1)
COMP = (CENTRAL2)                ; (3) MHD central (observation-2)

$PK

KA = THETA(1)*EXP(ETA(1))                    ; OXZ absorption

TVV2 = THETA(2)*(PKWT/50)**THETA(6)          ; OXZ central volume (scaling)
V2 = TVV2*EXP(ETA(2))                        ; OXZ central volume (individual)

TVV3 = THETA(3)*(PKWT/50)**THETA(6)          ; MHD central volume (scaling)
V3 = TVV3*EXP(ETA(3))                        ; MHD central volume (individual)

TVCL = THETA(4)*(PKWT/50)**THETA(7)          ; OXZ systemic CL (scaling)
CL = TVCL*EXP(ETA(4))                        ; OXZ systemic CL (individual)

TVCLM = THETA(5)*(PKWT/50)**THETA(8)         ; MHD systemic CL (scaling)
CLM = TVCLM*EXP(ETA(5))                      ; MHD systemic CL (individual)

kbt = THETA(9)*EXP(ETA(9))                   ; Backward (MHD to OXZ) rate constant

S2 = V2/1000                                 ; AMT (mg) to DV (ng/mL)
S3 = V3/1000                                 ; AMT (mg) to DV (ng/mL)

$DES
DADT(1)= - KA*A(1)
DADT(2)= KA*A(1) - (CL/V2)*A(2) + kbt*A(3)*0.992095          ; corrected for M.W. ratio
DADT(3)= (CL/V2)*A(2)*1.00796 - (CLM/V3)*A(3) - kbt*A(3)     ; corrected for M.W. ratio
```

```
; M.W. Ratio: OXZ/MHD=252.27/254.28=0.992095; MHD/OXZ= 254.28/252.27 = 1.00796

$ERROR
IPRED = F
IRES = DV - IPRED

IF(TYPE.EQ.1) THEN
Y=IPRED*(1+EPS(1))
ELSE
Y=IPRED*(1+EPS(2))
ENDIF

$THETA
        (0.1,0.2,)      ; KA (1)
        (20,50,)        ; V2 (2)
        50 FIX          ; V3 (3)
        (150,170,)      ; CL (4)
        (2,2.8,5)       ; CLM (5)
        (0.5,0.75,)     ; Exp-V (6)
        1 FIX           ; Exp-CL (7)
        (0.5,0.75,)     ; Exp-CLM (8)
        (0.01,0.05 ,)   ; Kbt (9)

$OMEGA
        0 FIX       ; ETA(1)
        0 FIX       ; ETA(2)
        0.09        ; ETA(3)
        0.09        ; ETA(4)
        0.09        ; ETA(5)
        0 FIX       ; ETA(6)
        0 FIX       ; ETA(7)
        0 FIX       ; ETA(8)
        0 FIX       ; ETA(9)

$SIGMA
        0.09        ; proportional error (TYPE=1)
        0.09        ; proportional error (TYPE=2)

$ESTIMATION METHOD=1 INTERACTION MAXEVAL=9999 PRINT=5 SIGDIGITS=3 SIGL=10
NOABORT NOTBT NOOBT NOSBT SORT MSFO=J40.msf
$COVARIANCE COMPRESS PRINT=E UNCONDITIONAL

$TABLE STUDY SITENUM ID EVID DOSN AMT CMT DRTE DFRM TYPE DV IPRED IRES
CWRES MDV TAFD TALD SEX ETHN RACE RACEWB WTKG PKWT HTCM AGRP PNA
PMADY PMAWK OBESE BMI PBMI SCR CLCR BUN AST ALT ALB PROT KETO CARB VERA
PHENOB VALP TOPR LEVE PHENYT FOSPHENYT KA CL V2 CLM V3 ETA3 ETA4 ETA5
NOPRINT ONEHEADER FILE=M6.TAB
```